NASA CONTRACTOR REPORT

NASA CR-1937

THE CALCULATION OF THE EIGENVALUES AND EIGENFUNCTIONS OF MATHIEU'S EQUATION

by D. B. Hodge

Prepared by
THE OHIO STATE UNIVERSITY
Columbus, Ohio 43212
for Langley Research Center

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • JANUARY 1972

1. Report No. NASA CR-1937	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle THE CALCULATION OF THE EIGENVALUES AND EIGENFUNCTIONS OF MATHIEU'S EQUATION		5. Report Date January 1972
		6. Performing Organization Code
7. Author(s) D. B. Hodge		8. Performing Organization Report No. 2902-4
		10. Work Unit No.
9. Performing Organization Name and Address The Ohio State University ElectroScience Laboratory Department of Electrical Engineering Columbus, OH 43212		11. Contract or Grant No. NGL 36-008-138
		13. Type of Report and Period Covered Contractor Report
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546		14. Sponsoring Agency Code
15. Supplementary Notes		

16. Abstract

The eigenfunctions of Mathieu's equation are expanded in trigonometric series, and the resulting eigenvalue problem is cast in matrix form. This matrix is found to be a symmetric, triagonal matrix, and the eigenvalues are computed using the bisection method. The eigenfunction expansion coefficients are obtained by the standard recursion method. This computational technique for the eignevalues and eigenfunctions of Mathieu's equation is both rapid and accurate.

17. Key Words (Suggested by Author(s)) Mathematics; Computer program; Mathieu's Equation		18. Distribution Statement Unclassified - Unlimited	
19. Security Classif. (of this report) Unclassified	20. Security Classif. (of this page) Unclassified	21. No. of Pages 15	22. Price* $3.00

*For sale by the National Technical Information Service, Springfield, Virginia 22151

TABLE OF CONTENTS

THE CALCULATION OF THE EIGENVALUES AND EIGENFUNCTIONS OF MATHIEU'S EQUATION

I. INTRODUCTION

Mathieu's equation, Eq. (1), arises when the scalar Helmholtz equation is solved in the elliptic cylinder coordinate system by means of separation of variables.

$$(1)\qquad (1 - z^2)\frac{d^2f}{dz^2} - z\frac{df}{dz} + (a + 2q - 4qz^2)\, f = 0$$

Here the independent variable, z, may be related to either the angular or the radial elliptic variable, q is related to the ellipticity of the coordinate system, and a is the separation constant or eigenvalue (also often refered to as the characteristic value). The solutions of this equation have been discussed extensively in the literature, and a summary of this literature may be found in Reference 1. The elliptic coordinate system is also discussed in detail in Reference 2.

In the following a simple, direct method for the calculation of the eigenvalues and eigenfunctions of this equation is developed. The computational procedure is both rapid and accurate. This method is quite similar to a technique presented earlier[3] for the calculation of the eigenvalues and eigenfunctions of the oblate and prolate spheroidal wave equations.

II. THE EIGENVALUES

The r^{th} solution of Eq. (1) may be expanded in terms of the trigonometric functions.

$$(2)\qquad f_r(q,z) = \frac{A_0^r(q)}{\sqrt{2}} + \sum_{m=1}^{\infty} [A_m^r(q)\cos(mz) + B_m^r(q)\sin(mz)]$$

Note that the normalization of the leading expansion coefficient differs slightly from that customarily found in the literature. If this expansion is substituted into Eq. (1) four independent sets of recursion relations are obtained, Eqs. (3-11)

Case 1:

(3) $$a\,A_0 - \sqrt{2}\,q\,A_2 = 0$$

(4) $$-\sqrt{2}\,q\,A_0 + (a-2^2)\,A_2 - q\,A_4 = 0$$

(5) $$-q\,A_{m-2} + (a-m^2)\,A_m - q\,A_{m+2} = 0,\ m = 4, 6, 8, \cdots$$

Case 2:

(6) $$(a-1^2-q)\,A_1 - q\,A_3 = 0$$

(7) $$-q\,A_{m-2} + (a-m^2)\,A_m - q\,A_{m+2} = 0,\ m = 3, 5, 7, \cdots$$

Case 3:

(8) $$(a-2^2)\,B_2 - q\,B_4 = 0$$

(9) $$-q\,B_{m-2} + (a-m^2)\,B_m - q\,B_{m+2} = 0,\ m = 4, 6, 8, \cdots$$

Case 4:

(10) $$(a-1^2+q)\,B_1 - q\,B_3 = 0$$

(11) $$-q\,B_{m-2} + (a-m^2)\,B_m - q\,B_{m+2} = 0,\ m = 3, 5, 7, \cdots$$

Here both the eigenvalue, a, and the expansion coefficients, A_m and B_m, are dependent upon the order, r, of the solution; however, this dependency has been temporarily suppressed in the notation for the sake of simplicity. The choice of the particular set of recursion relations to be used is dependent upon the symmetry and periodicity of the desired solution. It is evident that the recursion relations for even solutions will involve only the A_m^r coefficients while the recursion relations for odd solutions will involve only the B_m^r coefficients. Further, if the solution has π periodicity in z then only coefficients having even subscripts, m, will be utilized. Similarly, if the solution has 2π periodicity in z then only coefficients having odd subscripts, m, will be utilized. These conditions are summarized in Table I along with a shorthand notation which will be used to denote these cases. The integers s and t will be used to identify the periodicity and symmetry of the desired solution as indicated in Table I. In addition the letter D is used to denote the appropriate A or B coefficient as shown in Table I. The subscripts have been shifted so that D_0 is the leading coefficient in each case; this permits some simplification in subsequent programming.

		PERIODICITY	
		π $s = 0$	2π $s = 1$
SYMMETRY	EVEN $t=0$	$A^r_m \neq 0$, m even $D^r_m = A^r_m$	$A^r_m \neq 0$, m odd $D^r_m = A^r_{m+1}$
	ODD $t=1$	$B^r_m \neq 0$, m even $D^r_m = B^r_{m+2}$	$B^r_m \neq 0$, m odd $D^r_m = B^r_{m+1}$

TABLE I
SYMMETRY AND PERIODICITY CONDITIONS

The adoption of the notation described above allows the four sets of recursion relations to be written compactly as one statement,

(12) $$-X_{m-2}q\, D^r_{m-2}(q) + (a_r - W_m(q))\, D^r_m(q) - X_m q\, D^r_{m+2}(q) = 0,$$

for $m = 0, 2, 4, \cdots$

where

(13) $$X_m = \sqrt{2} \quad \text{if } s = 0,\ t = 0,\ m = 0$$
$$= 1 \quad \text{otherwise,}$$

(14) $$W_m = [m + s + 2\,t(1-s)]^2 + V_m,$$

and

$$
(15)\qquad V_m = \begin{cases} +q & \text{if } s = 1,\ t = 0,\ m = 0 \\ -q & \text{if } s = 1,\ t = 1,\ m = 0 \\ 0 & \text{otherwise.} \end{cases}
$$

The set of equations defined by Eq. (12) may be written in matrix form as shown in Eq. (16)

$$
(16)\qquad \begin{bmatrix} (a_r - W_o) & -X_o q & 0 & 0 & \cdots \\ -X_o q & (a_r - W_2) & -X_2 q & 0 & \\ 0 & -X_2 q & (a_r - W_4) & -X_4 q & \\ 0 & 0 & -X_4 q & (a_r - W_6) & \cdots \\ 0 & 0 & 0 & & \cdots \\ \vdots & & & & \cdots \end{bmatrix} \begin{bmatrix} D_o^r \\ D_2^r \\ D_4^r \\ D_6^r \\ \\ \\ \end{bmatrix} = 0
$$

The resulting real matrix is both symmetrical and tridiagonal. If this matrix is truncated to an N-by-N matrix, the bisection method[4] may be used to determine the eigenvalues, a_r, in a rapid, accurate manner. This procedure has the distinct advantage that the speed of convergence is known prior to computation, i.e., the uncertainty of the unknown eigenvalue decreases by a factor of 2 upon each iteration. Test calculations have indicated that the truncation of the matrix to a dimension only slightly larger than the order of the largest eigenvalue produces no significant error. The N eigenvalues determined in this manner are denoted by a_r, $r = s, s + 2, s + 4, \cdots, s + 2N-2$, when arranged in order of increasing algebraic value. The bisection procedure is initiated by noting that

$$
|a_r| \le \max_m \left[|X_{m-2}| + |W_m| + |X_m| \right].
$$

This procedure has been tested for various cases with $0.05 \leq q \leq 5.0$ and r_{max} up to 30. In all test cases the bisection procedure was terminated after obtaining five significant figures. A sample set of eigenvalues is given in Table II. The computed eigenvalues were found to agree with tabulated values.[1,5] The listing of the subroutine for the eigenvalues is given in Appendix A.

	a_r	
ORDER	Hodge	Abramowitz & Stegun
0	-0.5800066E+01	-0.580004602E+01
2	0.7449100E+01	0.744910974E+01
4	0.1709660E+02	
6	0.3636098E+02	
8	0.6419890E+02	
10	0.1001261E+03	0.10012636922E+03
12	0.1440878E+03	
14	0.1960645E+03	
16	0.2560488E+03	
18	0.3240378E+03	
20	0.4000327E+03	
22	0.4840264E+03	
24	0.5760225E+03	
26	0.6760208E+03	
28	0.7840146E+03	

TABLE II
Sample set of eigenvalues for $q = 5.0$, even symmetry, and π periodicity. (5 significant figures)

III. THE EIGENFUNCTION EXPANSION COEFFICIENTS

Having obtained the eigenvalue, a_r, the eigenfunction expansion coefficient, D_m^r, are obtained readily by means of recursion. Since these coefficients reach their maximum value at $r \approx m$, the recursion is carried out in two directions. Equation (12) is used to recur upward until $m = r$,

$$D_{m+2}^r = \frac{(a_r - W_m)}{X_m\, q} D_m^r - \frac{X_{m-2}}{X_m} D_{m-2}^r, \tag{17}$$

where

$$D_0^r = 1 \text{ and } D_{-2}^r = 0.$$

Similarly, Eq. (12) is used to recur downward from $m = m_{max}$ to $m = r$,

$$\text{(18)} \qquad D_{m-2}^r = \frac{(a_r - W_m)}{X_{m-2}\, q} D_m^r - \frac{X_m}{X_{m-2}} D_{m+2}^r,$$

where $D_{m_{max}} \cong 10^{-30}$, $D_{m_{max+2}} \cong 0$, and m_{max} is taken to be significantly larger than r. Values of m_{max} as small as r + 5 have been used successfully; however, this choice will depend upon q and the accuracy desired. The set of coefficients for $r > m$ are then normalized such that the two sets agree for $m = r$ in order to obtain one consistent set of expansion coefficients.

Finally the normalization condition

$$\text{(19)} \qquad \int_0^{\pi} f^2 (z,q)\, dz = \pi/2$$

is imposed. Substituting Eq. (2) into Eq. (19) yields

$$\text{(20)} \qquad \sum_{m=0}^{\infty} D_m^2 = 1.$$

Eigenfunction expansion coefficients obtained in this manner have been computed for the same ranges as the eigenvalues discussed earlier with m_{max} up to 40. The agreement with tabulated values in all cases where possible was excellent. The listing of the subroutine for the eigenfunction expansion coefficient is given in Appendix B.

The program utilizing these procedures required a total compilation and execution time of 12 seconds on an IBM 360-75 computer for the calculation of 60 eigenvalues and 1,260 eigenfunction expansion coefficients.

IV. CONCLUSION

The eigenvalue problem associated with Mathieu's equation when cast in matrix form was found to yield a real, symmetric, tridiagonal matrix. Thus the highly efficient and accurate bisection method may be used immediately to determine the eigenvalues. The eigenfunction expansion coefficients are subsequently obtained by a standard recursive technique. This procedure was tested and the results were compared with previously tabulated results showing excellent agreement. No computational difficulties were encountered.

REFERENCES

1. Abramowitz, M. and Stegun, I.A., "Handbook of Mathematical Functions," p. 722, National Bureau of Standards, Applied Mathematics Series No. 55, June 1964.

2. Morse, P.M. and Feshbach, H., Methods of Theoretical Physics, Part I, Chapter 5, McGraw-Hill, 1953.

3. Hodge, D.B., "Eigenvalues and Eigenfunctions of the Spheroidal Wave Equation," J. Math. Phys., Vol. 11, p. 2308, August 1970.

4. Wilkinson, J.H., Algebraic Eigenvalue Problem, p. 229, Clarendon Press, 1965.

5. Blanch, G. and Clemm, D.S., "Tables Relating to the Radial Mathieu Functions," Aeronautical Research Laboratories, Office of Aerospace Research, United States Air Force.

APPENDIX A
EIGENVALUE SUBROUTINE

The Fortran IV subroutine listed in Fig. A-1 may be used to compute the first NMX eigenvalues, $a_r(q)$, of Mathieu's equation. The input parameters are:

NMAX = 2*NMX
NMX = number of eigenvalues desired
Q = ellipticity
IS = integer 0 or 1 depending upon the periodicity as defined in Table I of the text
IT = integer 0 or 1 depending upon the symmetry as defined in Table I of the text.

The NMX computed eigenvalues are output in the one-dimensional array, EIG.

The number of significant figures in the computed eigenvalues is determined by the value of ACC which is defined near the beginning of the subroutine. In the case shown approximately five significant figures will be produced. If this accuracy is not obtained within 60 iterations the subroutine will output the error statement shown in statement 42. The maximum number of allowed iterations may be adjusted by altering the constant in the statement preceeding statement 41.

```
      SUBROUTINE MATEIG(NMAX,Q,IS,IT,EIG)
      DIMENSION P(50),IP(50),ALPHA(50),BETA(50),EIG(50)
      ACC=1.0E-05
      N=NMAX+2
      NI=N+1
      P(1)=1.0
      IP(1)=1
      X=1
      IF((IS.EQ.0).AND.(IT.EQ.0)) X=1.4142136
      V=0
      IF((IS.EQ.1).AND.(IT.EQ.0)) V=Q
      IF((IS.EQ.1).AND.(IT.EQ.1)) V=-Q
      DO 2 I=1,N
      ALPHA(I)=-(2*(I-1)+IS+2*IT*(1-IS))**2-V
      BETA(I+1)=-X*Q
      X=1
    2 V=0
      BETA(N+1)=0.0
      BO=ABS(ALPHA(1))+ABS(BETA(2))
      DO 3 I=2,N
      AO=ABS(BETA(I))+ABS(ALPHA(I))+ABS(BETA(I+1))
      BETA(I)=BETA(I)*BETA(I)
      IF(AO-BO)3,3,4
    4 BO=AO
    3 CONTINUE
      AO=-BO
  500 NMX=NMAX/2
      DO 20 K=1,NMX
      A=AO
      B=BO
      IERR=-1
   21 IIS=0
      CO=(A+B)/2.
      IF(CO)50,22,50
   50 ERR=(B-A)/ABS(CO)
      IERR=IERR+1
      IF(IERR-60)40,41,41
   41 WRITE(6,42)K
   42 FORMAT(36HOITERATIONS EXCEEDED FOR EIGENVALUE ,I3)
      GO TO 700
   40 IF(ERR-ACC)24,24,22
   22 P(2)=ALPHA(1)-CO
      P(1)=1.0
      DO 5 I=3,NI
      ABP=ABS(P(I-1))
      IF(ABP.LT.10.0) GO TO 5
      P(I-2)=P(I-2)/ABP
      P(I-1)=P(I-1)/ABP
```

Fig. A-1. Eigenvalue subroutine.

```
  5 P(I)=(ALPHA(I-1)-CO)*P(I-1)-BETA(I-1)*P(I-2)
    DO 6 I=2,NI
    IF(P(I))14,8,9
  8 IF(P(I-1))9,9,14
 14 IP(I)=-1
    GO TO 10
  9 IP(I)=1
 10 IF(IP(I)-IP(I-1))6,11,6
 11 IIS=IIS+1
  6 CONTINUE
    IF(IIS-K)16,15,15
 15 A=CO
    GO TO 21
 16 B=CO
    GO TO 21
 24 BO=CO
700 EIG(K)=-CO
 20 CONTINUE
801 RETURN
    END
```

Fig. A-1. (Contd.)

APPENDIX B
EXPANSION COEFFICIENT SUBROUTINE

The Fortran IV subroutine for the eigenfunction expansion coefficients is listed in Fig. B-1. In addition to the parameters defined in Appendix A, this subroutine requires one additional input parameter, MAXR, which is the number of terms to be retained in the expansion. The expansion coefficients are output in the two-dimensional array, D(N,J), where N denotes the N^{th} eigenfunction and J denotes the J^{th} expansion coefficient.

```
      SUBROUTINE MATCOF(NMAX,Q,IS,IT,MAXR,EIG,D)
      DIMENSION EIG(50),D(50,50),DP(50)
      NMX=NMAX/2
      DO 1 N=1,NMX
      DP(MAXR+3)=0.
      DP(MAXR+2)=1.0E-30
      D(N,1)=0.
      D(N,2)=1.0
      X=1
      Y=1
      IF((IS.EQ.0).AND.(IT.EQ.0)) X=1.4142136
      V=0
      IF ((IS.EQ.1).AND.(IT.EQ.0)) V=Q
      IF ((IS.EQ.1).AND.(IT.EQ.1)) V=-Q
      DO 107 LL=1,MAXR
      L=LL
      IF(LL.GT.N) GO TO 300
      D(N,L+2)=-(Y*D(N,L))/X+(EIG(N)-(2*(L-1)+IS+2*IT*(1-IS))
     1**2-V)*D(N,L+1)/(X*Q)
      Y=X
      X=1
      V=0
      GO TO 107
  300 L=MAXR+N-LL+1
      DP(L+1)=-DP(L+3)+(EIG(N)-(2*L+IS+2*IT*(1-IS))**2-V)
     1*DP(L+2)/Q
  107 CONTINUE
      CON=D(N,N+2)/DP(N+2)
      DO 118 J=N,MAXR
  118 D(N,J+2)=CON*DP(J+2)
      SUM=0
      MRX=MAXR+2
      DO 301 J=2,MRX
  301 SUM=SUM+D(N,J)**2
      ALF=SQRT(SUM)
      DO 302 J=2,MRX
      D(N,J-1)=D(N,J)/ALF
  302 CONTINUE
    1 CONTINUE
      RETURN
      END
```

Fig. B-1. Eigenfunction expansion coefficient subroutine.